

Polynomial Universes and Dependent Types

C.B. Aberlé

David I. Spivak

Abstract

Awodey, later with Newstead, showed how polynomial pseudomonads $(u, 1, \Sigma)$ with extra structure (termed "natural models" by Awodey) hold within them the syntax and rules for dependent type theory. Their work presented these ideas clearly but ultimately led them outside of the category of polynomial functors in order to explain all of the structure possessed by such models of type theory.

This paper builds off that work—explicating the syntax and rules for dependent type theory by axiomatizing them *entirely* in the language of polynomial functors. In order to handle the higher-categorical coherences required for such an explanation, we work with polynomial functors internally in the language of Homotopy Type Theory, which allows for higher-dimensional structures such as pseudomonads, etc. to be expressed purely in terms of the structure of a suitably-chosen ∞ -category of polynomial functors. The move from set theory to Homotopy Type Theory thus has a twofold effect of enabling a simpler exposition of natural models, which is at the same time amenable to formalization in a proof assistant, such as Agda.

Moreover, the choice to remain firmly within the setting of polynomial functors reveals many additional structures of natural models that were otherwise left implicit or not considered by Awodey & Newstead. Chief among these, we highlight the fact that every polynomial pseudomonad $(u, 1, \Sigma)$ as above that is also equipped with structure to interpret dependent product types gives rise to a self-distributive law $u \triangleleft u \rightarrow u \triangleleft u$, which witnesses the usual distributive law of dependent products over dependent sums.

1 Introduction

Dependent type theory [Mar75] was founded by Per Martin-Löf in 1975 (among others) to formalize constructive mathematics. The basic idea is that the *order of events* is fundamental to the mathematical story arc: when playing out any specific example story in that arc, the beginning of the story affects not only the later events, but even the very terms with which the later events will be described. For example, in the story arc of conditional probability, one may say “now if the set P that we are asked to condition on happens to have measure zero, we must stop; but assuming that’s not the case then the result will be a new probability measure.” Here the story teller is saying that no terms will describe what happens if P has measure zero, whereas otherwise the terms of standard probability will apply.

Dependent types form a logical system with syntax, rules of computation, and methods of deduction. In [Awo14; AN18], Awodey and later Newstead show that there is a strong connection between dependent type theory and polynomial functors, via their concept of *natural models*, which cleanly solve the problem of *strictifying* certain identities that typically hold only up to isomorphism in arbitrary categories, but must hold *strictly* in order for these to soundly model dependent type theory. The solution to this problem offered by Awodey and Newstead makes use of the type-theoretic concept of a *universe*. Such universes then turn out to naturally be regarded as polynomial functors on a suitably-chosen category of presheaves, satisfying a certain *representability* condition.

Although the elementary structure of natural models is thus straightforwardly described by considering them as objects in the category of polynomial functors, Awodey and Newstead were ultimately led outside of this category in order to fully explicate those parts of natural models that require identities to hold only *up to isomorphism*, rather than strictly. There is thus an evident tension between *strict* and *weak* identities that has not yet been fully resolved in the story of natural models. In the present work, we build on Awodey and Newstead’s work to fully resolve this impasse by showing how type universes can be fully axiomatized in terms of polynomial functors, by working with polynomial functors internally in the language of *Homotopy Type Theory* (HoTT). We thus come full circle from Awodey’s original motivation to develop natural models of Homotopy Type Theory, to describing natural models *in* Homotopy Type Theory.

The ability for us to tell the story of natural models as set entirely ◇ David says: Is that true? ◇ in the category of polynomial functors has a great simplifying effect upon the resultant theory, and reveals many additional structures, both of polynomial universes, and of the category of polynomial functors as a whole. As an illustration of this, we show how every polynomial universe u , regarded as a polynomial pseudomonad with additional structure, gives rise to self-distributive law $u \triangleleft u \rightarrow u \triangleleft u$, which in particular witnesses the usual distributive law of dependent products over dependent sums.

Moreover, the move from set theory to HoTT as a setting in which to tell this story enables new tools to be applied for its telling. In particular, the account of polynomial universes we develop is well-suited to formalization in a proof assistant, and we present such a formalization in Agda. This paper is thus itself a literate Agda document in which all results have been fully formalized and checked for validity.

```
{-# OPTIONS --without-K --rewriting #-}
module poly-universes where

open import Agda.Primitive
open import Agda.Builtin.Sigma
open import Agda.Builtin.Unit
```

The structure of this paper is as follows: ◇ David says: Fill ◇

2 Background on Type Theory, Natural Models & HoTT

We begin with a recap of natural models, dependent type theory, and HoTT, taking this also as an opportunity to introduce the basics of our Agda formalization.

2.1 Dependent Types and their Categorical Semantics

The question “what is a type” is as deep as it is philosophically fraught. For present purposes, however, we need not concern ourselves so much directly with what (dependent) type *are*, as with what they can *do*, and how best to mathematically model this behavior. Suffice it to say, then, that a type specifies rules for both constructing and using the *inhabitants* of that type in arbitrary contexts of usage. Following standard conventions, we use the notation $a : A$ to mean that a is an inhabitant of type A .

In Agda, one example of such a type is the *unit type* \top , which is defined to have a single inhabitant $tt : \top$, such that for any other inhabitant $x : \top$ we must have $x = tt$.

Another type (or rather, family of types) of particular importance is the *universe* of types Type , whose inhabitants themselves represent types.¹ So e.g. to say that \top , as defined above, is a type, we may simply write $\top : \text{Type}$. ♦ David says: The agda below involves levels and sets, but they aren’t discussed above. Say something about levels? ♦

```
Type : (ℓ : Level) → Set (lsuc ℓ)
Type ℓ = Set ℓ
```

Given a type A , one may in turn consider families of types $B\ x$ indexed by, or *dependent* upon arbitrary inhabitants $x : A$. In agda, we represent such a type family B as a function $A \rightarrow \text{Type}$.

Given a type $A : \text{Type}$ and a family of types $B : A \rightarrow \text{Type}$ as above, two key examples of types we may construct are:

- The *dependent function type* $(x : A) \rightarrow B\ x$, whose inhabitants are functions $\lambda\ x \rightarrow f\ x$ such that, for all $a : A$, we have $f\ a : B\ a$.
- The *dependent pair type* $\Sigma\ A\ B$, whose inhabitants are of the form $(a\ ,\ b)$ for $a : A$ and $b : B\ a$, such that there are functions $\text{fst} : \Sigma\ A\ B \rightarrow A$ and $\text{snd} : (p : \Sigma\ A\ B) \rightarrow B\ (\text{fst}\ p)$.

Note that in the case where B does not depend upon $x : A$ (i.e. the variable x does not appear in the expression for B), these correspond to the more familiar function type $A \rightarrow B$ and pair type $A \times B$, respectively. E.g. we can define the Cartesian product of two types A and B as follows:

```
_×_ : ∀ {ℓ κ} (A : Type ℓ) (B : Type κ) → Type (ℓ ⊔ κ)
A × B = Σ A (λ _ → B)
```

¹For consistency with the usage of the term “set” in HoTT (whereby sets are types satisfying a certain *truncation* condition, to be explained shortly,) we relabel Agda’s universes of types as Type , rather than the default Set .

In more traditional type-theoretic notation, one might see the rules for these types written as follows:

$$\begin{array}{c}
\Gamma \vdash \top : \text{Type} \quad \Gamma \vdash \text{tt} : \top \quad \frac{\Gamma \vdash x : \top}{\Gamma \vdash x = \text{tt}} \\[10pt]
\frac{\Gamma \vdash A : \text{Type} \quad \Gamma, x : A \vdash B[x] : \text{Type}}{\Gamma \vdash \Pi x : A. B[x] : \text{Type}} \quad \frac{\Gamma \vdash A : \text{Type} \quad \Gamma, x : A \vdash B[x] : \text{Type}}{\Gamma \vdash \Sigma x : A. B[x] : \text{Type}} \\[10pt]
\frac{\Gamma, x : A \vdash f[x] : B[x]}{\Gamma \vdash \lambda x. f[x] : \Pi x : A. B[x]} \quad \frac{\Gamma \vdash a : A \quad \Gamma \vdash b : B[a]}{\Gamma \vdash (a, b) : \Sigma x : A. B[x]} \\[10pt]
\frac{\Gamma \vdash f : \Pi x : A. B[x] \quad \Gamma \vdash a : A}{\Gamma \vdash fa : B[a]} \quad \frac{\Gamma \vdash p : \Sigma x : A. B[x]}{\Gamma \vdash \pi_1(p) : A} \quad \frac{\Gamma \vdash p : \Sigma x : A. B[x]}{\Gamma \vdash \pi_2(p) : B[\pi_1(p)]} \\[10pt]
(\lambda x. f[x])a = f[a] \quad \pi_1(a, b) = a \quad \pi_2(a, b) = b \\[10pt]
f = \lambda x. fx \quad p = (\pi_1(p), \pi_2(p))
\end{array}$$

The constructors λ and $(-, -)$ are called the *introduction* forms of $\Pi x : A. B[x]$ and $\Sigma x : A. B[x]$, while fa and $\pi_1(p)$, $\pi_2(p)$ are called the *elimination* forms of these types, respectively. One may wonder why all typing judgments in the above rules have been decorated with annotations of the form $\Gamma \vdash$, for some Γ . In these cases, Γ is the *context* of the corresponding judgment, used to keep track of the types of variables that may appear in that judgment.

Although contexts may seem rather trivial from a syntactic perspective, they are key to understanding the categorical semantics of dependent type theory. In particular, when modelling a dependent type theory as a category, it is the *contexts* which form the objects of this category, with morphisms between contexts being *substitutions* of terms in the domain context for the variables of the codomain context. A type A dependent upon variables in a context Γ is then interpreted as a morphism (i.e. substitution) $\Gamma, x : A \rightarrow \Gamma$, whose domain represents the context Γ extended with a variable of type A . We then interpret a term a of type A in context Γ as a *section* of the display map representing A , i.e.

$$\begin{array}{ccc}
\Gamma & \xrightarrow{a} & \Gamma, x : A \\
& \searrow & \downarrow A \\
& & \Gamma
\end{array}$$

Hence for each context Γ , there is a category $\mathbf{Ty}[\Gamma]$, which is the full subcategory of the slice category \mathcal{C}/Γ consisting of all display maps, wherein objects correspond to types in context Γ , and morphisms correspond to terms.

In typical categorical semantics, given a substitution $f : \Gamma \rightarrow \Delta$, and a type $A : \Delta, x : A \rightarrow \Delta$, we then interpret the action of f on A as a pullback:

$$\begin{array}{ccc}
\Gamma, x : A[f] & \longrightarrow & \Delta, x : A \\
A[f] \downarrow & \lrcorner & \downarrow A \\
\Gamma & \xrightarrow{f} & \Delta
\end{array}$$

In particular, then, any display map $A : \Gamma, x : A \rightarrow \Gamma$ induces a functor $\mathbf{Ty}[\Gamma] \rightarrow \mathbf{Ty}[\Gamma, x : A]$ by substitution along A . The left and right adjoints to this functor (if they exist) then correspond to dependent pair and dependent function types, respectively.

So far, we have told a pleasingly straightforward story of how to interpret the syntax of dependent type theory categorically. Unfortunately, this story is a fantasy, and the interpretation of type-theoretic syntax into categorical semantics sketched above is unsound, as it stands. The problem in essentials is that, in the syntax of type theory, substitution is strictly associative – i.e. given substitutions $f : \Gamma \rightarrow \Delta$ and $g : \Delta \rightarrow \Theta$ and a type A , we have $A[g][f] = A[g[f]]$; however, in the above categorical semantics, such iterated substitution is interpreted via successively taking pullbacks, which is in general only associative up to isomorphism. It seems, then, that something more is needed to account for this kind of *strictness* in the semantics of dependent type theory.² It is precisely this problem which natural models exist to solve.

2.2 Natural Models

The key insight of Awodey in formulating the notion of a natural model is that the problem of strictness in the semantics of type theory has, in a sense, already been solved by the notion of *type universes*, such as \mathbf{Type} as introduced above. Given a universe of types \mathcal{U} , rather than representing dependent types as display maps, and substitution as pullback, we can simply represent a family of types $B[x]$ dependent upon a type A as a function $A \rightarrow \mathcal{U}$, with substitution then given by precomposition, which is automatically strictly associative.

To interpret the syntax of dependent type theory in a category \mathcal{C} of contexts and substitutions, it therefore suffices to *embed* \mathcal{C} into a category whose type-theoretic internal language possesses such a universe whose types correspond to those of \mathcal{C} . For this purpose, we work in the category of *prehseaves* $\mathbf{Set}^{\mathcal{C}^{\text{op}}}$, with the embedding $\mathcal{C} \hookrightarrow \mathbf{Set}^{\mathcal{C}^{\text{op}}}$ being nothing other than the Yoneda embedding.

The universe \mathcal{U} is then given by an object of $\mathbf{Set}^{\mathcal{C}^{\text{op}}}$, i.e. an assignment, to each context $\Gamma : \text{Ob } \mathcal{C}$, of a set $\text{Ty}[\Gamma]$ of types in context Γ , with functions $\text{Ty}[\Delta] \rightarrow \text{Ty}[\Gamma]$ for each substitution $f : \Gamma \rightarrow \Delta$ that compose associatively, together with a \mathcal{U} -indexed family of objects $u \in \mathbf{Set}^{\mathcal{C}^{\text{op}}}/\mathcal{U}$, i.e. a natural transformation $u : \mathcal{U}_{\bullet} \Rightarrow \mathcal{U}$, where for each context Γ and type $A \in \text{Ty}[\Gamma]$, the fibre of u_{Γ} over A is the set $\text{Tm}[\Gamma, A]$ of inhabitants of A in context Γ .

The condition that all types in \mathcal{U} “belong to \mathcal{C} ”, in an appropriate sense, can then be expressed by requiring u to be *representable*, i.e. for any representable $\gamma \in \mathbf{Set}^{\mathcal{C}^{\text{op}}}$ with a natural transformation $\alpha : \gamma \Rightarrow \mathcal{U}$, the pullback

$$\begin{array}{ccc} \gamma \times_{\alpha, u} \mathcal{U}_{\bullet} & \Longrightarrow & \mathcal{U}_{\bullet} \\ u[\alpha] \downarrow & \lrcorner & \downarrow u \\ \gamma & \xrightarrow{\alpha} & \mathcal{U} \end{array}$$

²Something something Beck-Chevalley. . .

of u along α is representable. \diamond **David says:** I've always been confused by this. If $\mathcal{C} = 1$, then there is only one representable functor, the terminal set. So doesn't this condition say that $\mathcal{U}_\bullet = \mathcal{U}$? But this is not the universe for set-polynomials. \diamond

The question, then, is how to express that \mathcal{C} has dependent pair types, dependent function types, etc., in terms of the structure of u . A further insight of Awodey, toward answering this question, is that u gives rise to a functor (indeed, a *polynomial functor*) $\bar{u} : \mathbf{Set}^{\mathcal{C}^{\text{op}}} \rightarrow \mathbf{Set}^{\mathcal{C}^{\text{op}}}$, defined on $P : \mathbf{C}^{\text{op}} \rightarrow \mathbf{Set}$ as follows

$$\bar{u}(P)(\Gamma) = \sum_{A : \text{Ty}[\Gamma]} P(\Gamma)^{\text{Trm}[\Gamma, A]}$$

and much of the type-theoretic structure of u can be accounted for in terms of this functor. For instance (for reasons to be explained shortly), dependent pair types are given by a natural transformation $\sigma : \bar{u} \circ \bar{u} \Rightarrow \bar{u}$, that is *Cartesian* in that, for every $\alpha : P \Rightarrow Q$, the following naturality square is a pullback

$$\begin{array}{ccc} \bar{u}(\bar{u}(P)) & \xrightarrow{\bar{u}(\bar{u}(\alpha))} & \bar{u}(\bar{u}(Q)) \\ \sigma_P \downarrow & \lrcorner & \downarrow \sigma_Q \\ \bar{u}(P) & \xrightarrow{\bar{u}(\alpha)} & \bar{u}(Q) \end{array}$$

\diamond **David says:** Say somewhere that a map of polynomials is cartesian iff for each position, the map on directions is an iso? That's certainly how I think about this. \diamond

A question that arises, then, is what structure such a natural transformation interpreting dependent pair types must possess. It is natural to think that σ , along with a suitably-chosen natural transformation $\text{Id} \Rightarrow \bar{u}$, ought to give \bar{u} the structure of a monad. However, this turns out to be too strong a requirement, as it amounts to asking that $\Sigma x : A. (\Sigma y : B[x]. C[x, y]) = \Sigma(x, y) : (\Sigma x : A. B[x]). C[x, y]$, when in general this identity only holds up to isomorphism. Hence we seem to have crossed over from Scylla of our semantics for dependent type theory not being strict enough to interpret those identities we expect to hold strictly, to the Charybdis of them now being too strict to interpret the identities we expect to hold only up to isomorphism. It was for this reason that Awodey & Newstead were forced to ultimately go beyond Polynomial functors in their accounts of natural models.

However, another possibility exists to solve this dilemma – to use the language of HoTT itself to reason about such equality-up-to-isomorphism in natural models. For this purpose, rather than taking natural models to be certain (representable) morphisms in $\mathbf{Set}^{\mathcal{C}^{\text{op}}}$, we can instead expand the mathematical universe in which these models live to $\infty\mathbf{Grpd}^{\mathcal{C}^{\text{op}}}$, which, as an ∞ -topos, has HoTT as its internal language. Taking advantage of this fact, we can use HoTT itself as a language for studying the semantics of type theory, by postulating an abstract type \mathcal{U} together with a type family $u : \mathcal{U} \rightarrow \text{Type}$, corresponding to a representable natural transformation $u : \mathcal{U}_\bullet \Rightarrow \mathcal{U}$ as above.

What remains, then, is to show how the various type-theoretic properties of such natural models can be expressed in terms of polynomial functors in the language of HoTT, and the complex identities to which these give rise. For this purpose, we begin with a recap of the basics of HoTT, before launching into a development of the theory of

polynomial functors within HoTT, with an eye toward the latter's use in the study of natural models.

2.3 Homotopy Type Theory

2.3.1 The Identity Type

Given elements $a, b : A$ for some type A , the identity type $a \equiv b$ is inductively generated from the single constructor $\text{refl} : \{x : A\} \rightarrow x \equiv x$, witnessing reflexivity of equality.

```
open import Agda.Builtin.Equality
open import Agda.Builtin.Equality.Rewrite
```

The core insight of Homotopy Type Theory is that the presence of (intensional) identity types in a system of dependent type theory endows each type with the structure of an ∞ -groupoid, and endows each function between types with the structure of a functor between ∞ -groupoids, etc. This allows a wealth of higher-categorical properties and structures to be defined and studied *internally* in the language of dependent type theory.

Since an invocation of reflexivity typically occurs at the end of an equality proof, we introduce the notation \square as a shorthand for refl as follows:

```
_□_ : ∀ {ℓ} {A : Type ℓ} (a : A) → a ≡ a
a □ = refl
```

The inductive generation of $a \equiv b$ from refl then gives rise to the operation of *transport* that allows an inhabitant of the type $B\ a$ to be converted to an inhabitant of $B\ b$ for any type family $B : (x : A) \rightarrow \text{Type}$.

```
transp : ∀ {ℓ κ} {A : Type ℓ} (B : A → Type κ) {a a' : A}
         → (e : a ≡ a') → B a → B a'
transp B refl b = b
```

Transitivity of equality then follows in the usual way. Note, however, that we take advantage of Agda's support for infix notation to present transitivity in such a way as to streamline both the reading and writing of equality proofs:

```
_≡⟨_⟩_ : ∀ {ℓ} {A : Type ℓ} (a : A) {b c : A}
         → a ≡ b → b ≡ c → a ≡ c
a ≡⟨ e ⟩ refl = e
```

Symmetry of equality follows similarly:

```
sym : ∀ {ℓ} {A : Type ℓ} {a a' : A} → a ≡ a' → a' ≡ a
sym refl = refl
```

As mentioned above, each function $f : A \rightarrow B$ in HoTT is canonically endowed with the structure of a functor between ∞ -groupoids, where the action of such a function f on paths (i.e. elements of the identity type) is as follows:

```
ap : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {a a' : A}
    → (f : A → B) → a ≡ a' → (f a) ≡ (f a')
ap f refl = refl
```

By the same token, given a proof $f \equiv g$ for two functions $f, g : (x : A) \rightarrow B$, it follows that for any $a : A$ we have $f a \equiv g a$.

```
coAp : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f g : A → B}
    → f ≡ g → (x : A) → f x ≡ g x
coAp refl x = refl
```

Moreover, to show that two pairs (a, b) and (a', b') are equal, it suffices to show that there is an identification $e : a \equiv a'$ together with $e' : \text{transp } B \ e \ b \equiv b'$.

```
pairEq : ∀ {ℓ κ} {A : Type ℓ} {B : A → Type κ}
    → {a a' : A} {b : B a} {b' : B a'}
    → (e : a ≡ a') (e' : transp B e b ≡ b')
    → (a, b) ≡ (a', b')
pairEq refl refl = refl
```

2.3.2 Truncation, Bracket Types & Factorization

We say that a type A is:

1. *contractible* (aka (-2) -truncated) if A is uniquely inhabited
2. a (mere) *proposition* (aka (-1) -truncated) if any two elements of A are identical
3. a *set* (aka 0 -truncated) if for any $a, b : A$, the type $a \equiv b$ is a proposition.

```
isContr : ∀ {ℓ} → Type ℓ → Type ℓ
isContr A = Σ A (λ a → (b : A) → a ≡ b)
```

```
isProp : ∀ {ℓ} → Type ℓ → Type ℓ
isProp A = {a b : A} → a ≡ b
```

```
isSet : ∀ {ℓ} → Type ℓ → Type ℓ
isSet A = (a b : A) → isProp (a ≡ b)
```

We additionally postulate the existence of a *propositional truncation*, or *bracket type* operation, that takes a type A to the least proposition (with respect to entailment) entailed by inhabitation of A .

postulate

```

||_|| : ∀ {ℓ} (A : Type ℓ) → Type lzero
in||-|| : ∀ {ℓ} {A : Type ℓ} → A → || A ||
||-||IsProp : ∀ {ℓ} {A : Type ℓ} → isProp (|| A ||)
||-||Rec : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ}
  → isProp B → (A → B) → || A || → B

```

When the type $\| A \|$ is inhabited, we say that A is *merely* inhabited. From this operation on types, we straightforwardly obtain the higher analogue of the usual epi-mono factorization system on functions between sets, as follows:

Given a function $f : A \rightarrow B$ and an element $b : B$, the *fibre* of f at b is the type of elements of A equipped with a proof of $f\ a \equiv b$:

```

Fibre : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → B → Type (ℓ ⊔ κ)
Fibre {A = A} f b = Σ A (λ a → f a ≡ b)

```

◇ David says: Why did we need $\{A = A\}$? ◇

By the same token, given such an f , its *image* is the type of elements of B whose fibres are merely inhabited.

```

Im : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → Type κ
Im {B = B} f = Σ B (λ b → || Fibre f b ||)

```

We say that f is *(-1)-truncated* (i.e. a monomorphism), if for each $b : B$, the fibre of f at b is a proposition.

```

isMono : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → Type (ℓ ⊔ κ)
isMono {B = B} f = (b : B) → isProp (Fibre f b)

```

Likewise, we say that f is *(-1)-connected* (i.e. an epimorphism), if all of its fibres are merely inhabited.

```

isEpi : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → Type κ
isEpi {B = B} f = (b : B) → || Fibre f b ||

```

Every function f can then be factored into a *(-1)-connected* map onto its image followed by a *(-1)-truncated* map onto its codomain:

```

factor1 : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (f : A → B) → A → Im f
factor1 f a = (f a) , in||-|| (a , refl)

```

```

factor2 : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (f : A → B) → Im f → B
factor2 f (b , _) = b

```

```

factor≡ : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ}
  → (f : A → B) (a : A) → factor2 f (factor1 f a) ≡ f a
factor≡ f a = refl

```

```

factor1IsEpi : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ}
  → (f : A → B) → isEpi (factor1 f)
factor1IsEpi f (b , x) =
  ||-||Rec ||-||IsProp
    (λ {(a , refl) → in||-|| (a , pairEq refl ||-||IsProp)})
    x

factor2IsMono : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ}
  → (f : A → B) → isMono (factor2 f)
factor2IsMono f b {a = ((b' , x) , refl)} {b = ((b'' , x') , refl)} =
  pairEq (pairEq refl ||-||IsProp) (lemma ||-||IsProp)
  where lemma : (e : x ≡ x') → transp (λ a → factor2 f a ≡ b)
    (pairEq refl e) refl ≡ refl
    lemma refl = refl

```

2.3.3 Equivalences

A pivotal notion, both for HoTT in general, and for the content of this paper, is that of a function $f : A \rightarrow B$ being an *equivalence* of types. The reader familiar with HoTT will know that there are several definitions – all equivalent – of this concept appearing in the HoTT literature. For present purposes, we make use of the *bi-invertible maps* notion of equivalence. Hence we say that a function $f : A \rightarrow B$ is an equivalence if it has both a left inverse and a right inverse:

```

isEquiv : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → Type (ℓ ⊔ κ)
isEquiv {A = A} {B = B} f =
  (Σ (B → A) (λ g → (a : A) → g (f a) ≡ a))
  × (Σ (B → A) (λ h → (b : B) → f (h b) ≡ b))

```

A closely-related notion is that of a function f being an *isomorphism*, i.e. having a single two-sided inverse:

```

Iso : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} → (A → B) → Type (ℓ ⊔ κ)
Iso {A = A} {B = B} f =
  (Σ (B → A) (λ g → ((a : A) → g (f a) ≡ a)
    × ((b : B) → f (g b) ≡ b)))

```

One might be inclined to wonder, then, why we bother to define equivalence via the seemingly more complicated notion of having both a left and a right inverse when the familiar notion of isomorphism can just as well be defined, as above. The full reasons for this are beyond the scope of this paper, though see [hottbook] for further discussion. Suffice it to say that, for subtle reasons due to the higher-categorical structure of types in HoTT, the plain notion of isomorphism given above is not a *good* notion of equivalence, whereas that of bi-invertible maps is. In particular, the type $\text{Iso } f$ is not necessarily a proposition for arbitrary f , whereas $\text{isEquiv } f$ is.

We may nonetheless move more-or-less freely back and forth between the notions of equivalence and isomorphism given above, thanks to the following functions, which allow us to convert isomorphisms to equivalences and vice versa:

```

Iso→isEquiv : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B}
              → Iso f → isEquiv f
Iso→isEquiv (g , l , r) = ((g , l) , (g , r))

isEquiv→Iso : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B}
              → isEquiv f → Iso f
isEquiv→Iso {f = f} ((g , l) , (h , r)) =
  h , (λ x → (h (f x))           ≡⟨ sym (l (h (f x))) ⟩
        (g (f (h (f x)))) ≡⟨ ap g (r (f x)) ⟩
        ((g (f x))           ≡⟨ l x ⟩
        (x □)))) , r

```

Straightforwardly, the identity function at each type is an equivalence, and equivalences are closed under composition:

```

idIsEquiv : ∀ {ℓ} {A : Type ℓ} → isEquiv {A = A} (λ x → x)
idIsEquiv = ((λ x → x) , (λ x → refl)) , ((λ x → x) , (λ x → refl))

compIsEquiv : ∀ {ℓ ℓ' ℓ''} {A : Type ℓ} {B : Type ℓ'} {C : Type ℓ''}
              → (g : B → C) (f : A → B) → isEquiv g → isEquiv f
              → isEquiv (λ a → g (f a))
compIsEquiv g f ((g' , lg) , (g'' , rg)) ((f' , lf) , (f'' , rf)) =
  ( (λ c → f' (g' c))
    , λ a → (f' (g' (g (f a)))) ≡⟨ ap f' (lg (f a)) ⟩
          (f' (f a))           ≡⟨ lf a ⟩
          (a                     □)))
  , ((λ c → f'' (g'' c))
     , λ c → (g (f (f'' (g'' c)))) ≡⟨ ap g (rf (g'' c)) ⟩
           (g (g'' c))           ≡⟨ rg c ⟩
           (c                     □)))

```

And by the above translation between equivalences and isomorphisms, each equivalence has a corresponding inverse map in the opposite direction, which is itself an equivalence:

```

inv : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B} → isEquiv f → B → A
inv (_, (h , _)) = h

isoInv : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B}
        → (isof : Iso f) → Iso (fst isof)
isoInv {f = f} (g , l , r) = (f , r , l)

```

```

invIsEquiv : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B}
  → (ef : isEquiv f) → isEquiv (inv ef)
invIsEquiv ef = Iso→isEquiv (isoInv (isEquiv→Iso ef))

```

We close this section by noting that, for each type family $B : A \rightarrow \text{Type}$, the map B $a \rightarrow B$ a' induced by transport along $e : a \equiv a'$ for any $a, a' : A$ is an equivalence with inverse given by transport along $\text{sym } e$, as follows:

```

sym1 : ∀ {ℓ κ} {A : Type ℓ} {B : A → Type κ} {a b : A}
  → (e : a ≡ b) (x : B a) → transp B (sym e) (transp B e x) ≡ x
sym1 refl x = refl

```

```

symr : ∀ {ℓ κ} {A : Type ℓ} {B : A → Type κ} {a b : A}
  → (e : a ≡ b) (y : B b) → transp B e (transp B (sym e) y) ≡ y
symr refl x = refl

```

And... ◇ David says: You probably plan to say more here? It's a lot of stuff... ◇

```

transpAp : ∀ {ℓ ℓ' κ} {A : Type ℓ} {A' : Type ℓ'} {a b : A}
  → (B : A' → Type κ) (f : A → A') (e : a ≡ b) (x : B (f a))
  → transp (λ x → B (f x)) e x ≡ transp B (ap f e) x
transpAp B f refl x = refl

```

```

≡sim1 : ∀ {ℓ} {A : Type ℓ} {a b : A}
  → (e : a ≡ b) → refl ≡ (b ≡⟨ sym e ⟩ e)
≡sim1 refl = refl

```

```

≡idr : ∀ {ℓ} {A : Type ℓ} {a b : A}
  → (e : a ≡ b) → e ≡ (a ≡⟨ refl ⟩ e)
≡idr refl = refl

```

```

conj : ∀ {ℓ} {A : Type ℓ} {a b c d : A}
  → (e1 : a ≡ b) (e2 : a ≡ c) (e3 : b ≡ d) (e4 : c ≡ d)
  → (a ≡⟨ e1 ⟩ e3) ≡ (a ≡⟨ e2 ⟩ e4)
  → e3 ≡ (b ≡⟨ sym e1 ⟩ (a ≡⟨ e2 ⟩ e4))
conj e1 e2 refl refl refl = ≡sim1 e1

```

```

nat : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f g : A → B} {a b : A}
  → (α : (x : A) → f x ≡ g x) (e : a ≡ b)
  → ((f a) ≡⟨ α a ⟩ (ap g e)) ≡ ((f a) ≡⟨ ap f e ⟩ (α b))
nat {a = a} α refl = ≡idr (α a)

```

```

cancel : ∀ {ℓ} {A : Type ℓ} {a b c : A}
  → (e1 e2 : a ≡ b) (e3 : b ≡ c)
  → (a ≡⟨ e1 ⟩ e3) ≡ (a ≡⟨ e2 ⟩ e3)

```

```

      → e1 ≡ e2
cancel e1 e2 refl refl = refl

apId : ∀ {ℓ} {A : Type ℓ} {a b : A}
      → (e : a ≡ b) → ap (λ x → x) e ≡ e
apId refl = refl

apComp : ∀ {ℓ ℓ' ℓ''} {A : Type ℓ} {B : Type ℓ'} {C : Type ℓ''} {a b : A}
      → (f : A → B) (g : B → C) (e : a ≡ b)
      → ap (λ x → g (f x)) e ≡ ap g (ap f e)
apComp f g refl = refl

apHtpy : ∀ {ℓ} {A : Type ℓ} {a : A}
      → (i : A → A) (α : (x : A) → i x ≡ x)
      → ap i (α a) ≡ α (i a)
apHtpy {a = a} i α =
  cancel (ap i (α a)) (α (i a)) (α a)
    ((i (i a) ≡⟨ ap i (α a) ⟩ α a)
     ≡⟨ sym (nat α (α a)) ⟩
     ((i (i a) ≡⟨ α (i a) ⟩ ap (λ z → z) (α a))
      ≡⟨ ap (λ e → i (i a) ≡⟨ α (i a) ⟩ e) (apId (α a)) ⟩
      ((i (i a) ≡⟨ α (i a) ⟩ α a) □)))

HAdj : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ}
      → (A → B) → Set (ℓ ⊔ κ)
HAdj {A = A} {B = B} f =
  Σ (B → A) (λ g →
    Σ ((x : A) → g (f x) ≡ x) (λ η →
      Σ ((y : B) → f (g y) ≡ y) (λ ε →
        (x : A) → ap f (η x) ≡ ε (f x))))

Iso→HAdj : ∀ {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B}
      → Iso f → HAdj f
Iso→HAdj {f = f} (g , η , ε) =
  g , (η
    , ( (λ y →
        f (g y) ≡⟨ sym (ε (f (g y))) ⟩
        (f (g (f (g y)))) ≡⟨ ap f (η (g y)) ⟩
        (f (g y) ≡⟨ ε y ⟩
        (y □))))
    , λ x → conj (ε (f (g (f x)))) (ap f (η (g (f x))))
      (ap f (η x)) (ε (f x))
      (((f (g (f (g (f x)))) ≡⟨ ε (f (g (f x))) ⟩ ap f (η x))
       ≡⟨ nat (λ z → ε (f z)) (η x) ⟩

```

$$\begin{aligned}
& (((f (g (f (g (f x)))) \equiv \langle \text{ap } (\lambda z \rightarrow f (g (f z))) (\eta x) \rangle \epsilon (f x))) \\
& \equiv \langle \text{ap } (\lambda e \rightarrow (f (g (f (g (f x)))) \equiv \langle e \rangle \epsilon (f x))) \\
& \quad ((\text{ap } (\lambda z \rightarrow f (g (f z))) (\eta x)) \\
& \quad \equiv \langle \text{apComp } (\lambda z \rightarrow g (f z)) f (\eta x) \rangle \\
& \quad ((\text{ap } f (\text{ap } (\lambda z \rightarrow g (f z)) (\eta x))) \\
& \quad \equiv \langle \text{ap } (\text{ap } f) (\text{apHtpy } (\lambda z \rightarrow g (f z)) \eta) \rangle \\
& \quad (\text{ap } f (\eta (g (f x))) \square))) \rangle \\
& (((f (g (f (g (f x)))) \equiv \langle \text{ap } f (\eta (g (f x))) \rangle \epsilon (f x))) \square))))
\end{aligned}$$

```

pairEquiv1 : ∀ {ℓ ℓ' κ} {A : Type ℓ} {A' : Type ℓ'} {B : A' → Type κ}
  → (f : A → A') → isEquiv f
  → isEquiv {A = Σ A (λ x → B (f x))} {B = Σ A' B}
    (λ (x , y) → (f x , y))

```

```

pairEquiv1 {A = A} {A' = A'} {B = B} f ef =
  Iso→isEquiv
    ( (λ (x , y) → (g x , transp B (sym (ε x)) y))
      , (λ (x , y) → pairEq (η x) (lemma x y))
      , λ (x , y) → pairEq (ε x) (symr (ε x) y) ) )

```

where

```

g : A' → A
g = fst (Iso→HAdj (isEquiv→Iso ef))
η : (x : A) → g (f x) ≡ x
η = fst (snd (Iso→HAdj (isEquiv→Iso ef)))
ε : (y : A') → f (g y) ≡ y
ε = fst (snd (snd (Iso→HAdj (isEquiv→Iso ef))))
ρ : (x : A) → ap f (η x) ≡ ε (f x)
ρ = snd (snd (snd (Iso→HAdj (isEquiv→Iso ef))))
lemma : (x : A) (y : B (f x))
  → transp (λ z → B (f z)) (η x)
    (transp B (sym (ε (f x))) y)
  ≡ y
lemma x y = (transp (λ z → B (f z)) (η x)
  (transp B (sym (ε (f x))) y))
  ≡⟨ transpAp B f (η x)
    (transp B (sym (ε (f x))) y) ⟩
  ( transp B (ap f (η x))
    (transp B (sym (ε (f x))) y)
  ≡⟨ ap (λ e → transp B e
    (transp B (sym (ε (f x))) y))
    (ρ x) ⟩
  ( (transp B (ε (f x))
    (transp B (sym (ε (f x))) y))
  ≡⟨ (symr (ε (f x)) y) ⟩
  (y □))

```

```

pairEquiv2 : ∀ {ℓ κ κ'} {A : Type ℓ} {B : A → Type κ} {B' : A → Type κ'}
  → (g : (x : A) → B x → B' x) → ((x : A) → isEquiv (g x))
  → isEquiv {A = Σ A B} {B = Σ A B'}
    (λ (x , y) → (x , g x y))

pairEquiv2 g eg =
  let isog = (λ x → isEquiv→Iso (eg x))
  in Iso→isEquiv ( (λ (x , y) → (x , fst (isog x) y))
    , ( (λ (x , y) →
      pairEq refl (fst (snd (isog x)) y))
    , λ (x , y) →
      pairEq refl (snd (snd (isog x)) y)))

pairEquiv : ∀ {ℓ ℓ' κ κ'} {A : Type ℓ} {A' : Type ℓ'}
  → {B : A → Type κ} {B' : A' → Type κ'}
  → (f : A → A') (g : (x : A) → B x → B' (f x))
  → isEquiv f → ((x : A) → isEquiv (g x))
  → isEquiv {A = Σ A B} {B = Σ A' B'}
    (λ (x , y) → (f x , g x y))

pairEquiv f g ef eg =
  compIsEquiv (λ (x , y) → (f x , y))
    (λ (x , y) → (x , g x y))
    (pairEquiv1 f ef)
    (pairEquiv2 g eg)

```

◇ David says: Part of me does think most of this should be in the main paper, rather than the appendix. But this very last part just feels long... Maybe with more text it could be included? ◇

3 Polynomials in HoTT

◇ David says: Introduce the section ◇

Remark 3.1. Since essentially all of the categorical structures treated in this paper will be infinite-dimensional, we shall generally omit the prefix “ $(\infty, 1)$ –” from our descriptions these structures. Hence hereafter “category” generally means $(\infty, 1)$ -category, “functor” means $(\infty, 1)$ -functor, etc. ◇

3.1 Basics

Let **Type** be the category of types and functions between them. Given a type A , let y^A denote the corresponding representable functor **Type** \rightarrow **Type**.

A *polynomial functor* is a coproduct of representable functors $\mathbf{Type} \rightarrow \mathbf{Type}$, i.e. an endofunctor on \mathbf{Type} of the form

$$\sum_{a:A} y^{B(a)}$$

for some type A and family of types $B : A \rightarrow \mathbf{Type}$. The data of a polynomial functor is thus uniquely determined by the choice of A and B . Hence we may represent such functors in Agda simply as pairs (A, B) of this form:

```
Poly : (ℓ κ : Level) → Type ((lsuc ℓ) ⊔ (lsuc κ))
Poly ℓ κ = Σ (Set ℓ) (λ A → A → Set κ)
```

♦ **David says:** I was surprised to see *Set* show up, especially in the base. I thought we were allowing an arbitrary type here? ♦

A basic example of such a polynomial functor is the identity functor y consisting of a single term of unit arity – hence represented by the pair $(\top, \lambda _ \rightarrow \top)$.

```
y : Poly lzero lzero
y = (⊤ , λ _ → ⊤)
```

The observant reader may note the striking similarity of the above-given formula for a polynomial functor and the endofunctor on $\mathbf{Set}^{\text{cop}}$ defined in the previous section on natural models. ♦ **David says:** Either here or there, let's say that \mathbf{Type} means small object in $\mathbf{Set}^{\text{cop}}$, or whatever is correct along those lines. ♦ Indeed, this is no accident, given a type U and a function $u : U \rightarrow \mathbf{Type}$ corresponding to a natural model as described previously, we obtain the corresponding polynomial $u : \mathbf{Poly}$ as the pair (U, u) . Hence we can study the structure of U and u in terms of u , and this, as we shall see, allows for some significant simplifications in the theory of natural models.

Given polynomial functors $p = \sum_{a:A} y^{B(a)}$ and $q = \sum_{c:C} y^{D(c)}$, a natural transformation $p \Rightarrow q$ is equivalently given by the data of a *forward* map $f : A \rightarrow B$ and a *backward* map $g : (a : A) \rightarrow D (f a) \rightarrow B a$, as can be seen from the following argument via Yoneda:

$$\begin{aligned} \int_{y \in \mathbf{Type}} (\sum_{a:A} y^{B(a)}) &\rightarrow \sum_{c:C} y^{D(c)} \\ \simeq \prod_{a:A} \int_{y \in \mathbf{Type}} y^{B(a)} &\rightarrow \sum_{c:C} y^{D(c)} \\ \simeq \prod_{a:A} \sum_{c:C} B(a)^{D(c)} & \\ \simeq \sum_{f:A \rightarrow C} \prod_{a:A} B(a)^{D(f(c))} & \end{aligned}$$

We use the notation $p \Leftrightarrow q$ to denote the type of natural transformations from p to q (aka *lenses* from p to q), which may be written in Agda as follows:

```
_↔_ : ∀ {ℓ ℓ' κ κ'} → Poly ℓ κ → Poly ℓ' κ' → Type (ℓ ⊔ ℓ' ⊔ κ ⊔ κ')
(A , B) ↔ (C , D) = Σ (A → C) (λ f → (a : A) → D (f a) → B a)
```

By application of function extensionality, we derive the following type for proofs of equality between lenses:


```

EqLens : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (p ⇔ q) → (p ⇔ q) → Set (ℓ ⊔ ℓ' ⊔ κ ⊔ κ')
EqLens (A , B) (C , D) (f , g) (f' , g') =
  (a : A) → Σ (f a ≡ f' a)
    (λ e → (b : D (f a)) → g a b ≡ g' a (transp D e b))

```

For each polynomial p , the corresponding *identity* lens is given by the following data:

```

id : ∀ {ℓ κ} (p : Poly ℓ κ) → p ⇔ p
id p = ( (λ a → a) , λ a b → b )

```

And given lenses $p ⇔ q$ and $q ⇔ r$, their composition may be computed as follows:

```

comp : ∀ {ℓ ℓ' ℓ'' κ κ' κ''}
  → (p : Poly ℓ κ) (q : Poly ℓ' κ') (r : Poly ℓ'' κ'')
  → p ⇔ q → q ⇔ r → p ⇔ r
comp p q r (f , g) (h , k) =
  ( (λ a → h (f a)) , λ a z → g a (k (f a) z) )

```

♦ [David says](#): Use $(f, f^\#)$ and $(g, g^\#)$ here, as below? ♦

Hence we have a category **Poly** of polynomial functors and lenses between them. Our goal, then, is to show how the type-theoretic structure of a natural model naturally arises from the structure of this category. In fact, **Poly** is replete with categorical structures of all kinds, of which we now mention but a few.

3.2 The Vertical-Cartesian Factorization System on Poly

We say that a lens $(f , f^\#) : (A , B) ⇔ (C , D)$ is *vertical* if $f : A → C$ is an equivalence, and Cartesian if for every $a : A$, the map $f^\# a : D[f a] → B a$ is an equivalence.

```

isVertical : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → p ⇔ q → Set (ℓ ⊔ ℓ')
isVertical p q (f , f^\#) = isEquiv f

```

```

isCartesian : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → p ⇔ q → Set (ℓ ⊔ κ ⊔ κ')
isCartesian (A , B) q (f , f^\#) = (a : A) → isEquiv (f^\# a)

```

Every lens $(A , B) ⇔ (C , D)$ can then be factored as a vertical lens followed by a Cartesian lens:

```

vertfactor : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → p ⇔ (fst p , λ x → snd q (fst f x))
vertfactor p q (f , f^\#) = (λ x → x) , (λ a x → f^\# a x)

```

```

vertfactorIsVert : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ)
  → (q : Poly ℓ' κ') (f : p ⇔ q)
  → isVertical p (fst p , λ x → snd q (fst f x))
    (vertfactor p q f)
vertfactorIsVert p q f = idIsEquiv

cartfactor : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → (fst p , λ x → snd q (fst f x)) ⇔ q
cartfactor p q (f , f#) = f , λ a x → x

cartfactorIsCart : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ)
  → (q : Poly ℓ' κ') (f : p ⇔ q)
  → isCartesian (fst p , λ x → snd q (fst f x)) q
    (cartfactor p q f)
cartfactorIsCart p q f x = idIsEquiv

vertcartfactor≡ : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ)
  → (q : Poly ℓ' κ') (f : p ⇔ q)
  → EqLens p q f
    (comp p (fst p , λ x → snd q (fst f x)) q
      (vertfactor p q f)
      (cartfactor p q f))
vertcartfactor≡ p q f a = refl , (λ b → refl)

```

Of these two classes of morphisms in **Poly**, it is *Cartesian* lenses that shall be of principal interest to us. If we view a polynomial $p = (A, B)$ as an A -indexed family of types, given by B , then given a lens $(f, f\#) : p \rightleftharpoons u$ to the universe, we can think of the map $f\# a : u (f a) \rightarrow B a$, as an *elimination form* for the type $u (f a)$, i.e. a way of *using* elements of the type $u (f a)$. If we then ask that $(f, f\#)$ is Cartesian, this implies that the type $u (f a)$ is completely characterized (up to equivalence) by this elimination form, and in this sense, that u *contains* the type $B a$, for all $a : A$.³ We can therefore use Cartesian lenses to detect which types are contained in a natural model u .

A further fact about Cartesian lenses is that they are closed under identity and composition, as a direct consequence of the closure of equivalences under identity and composition:

```

idCart : ∀ {ℓ κ} (p : Poly ℓ κ)
  → isCartesian p p (id p)
idCart p a = idIsEquiv

```

³Those familiar with type theory may recognize this practice of defining types in terms of their elimination forms as characteristic of so-called *negative* types (in opposition to *positive* types, which are characterized by their introduction forms). Indeed, there are good reasons for this, having to do with the fact that negative types are equivalently those whose universal property is given by a *representable* functor, rather than a *co-representable* functor, which reflects the fact that natural models are defined in terms of *presheaves* on a category of contexts, rather than *co-presheaves*.

```

compCartesian : ∀ {ℓ ℓ' ℓ'' κ κ' κ''}
  → (p : Poly ℓ κ) (q : Poly ℓ' κ') (r : Poly ℓ'' κ'')
  → (f : p ⇔ q) (g : q ⇔ r)
  → isCartesian p q f → isCartesian q r g
  → isCartesian p r (comp p q r f g)
compCartesian p q r f g cf cg a =
  compIsEquiv (snd f a) (snd g (fst f a)) (cf a) (cg (fst f a))

```

Hence there is a category **Poly**^{Cart} defined as the wide subcategory of **Poly** whose morphisms are precisely Cartesian lenses. As we shall see, much of the categorical structure of natural models qua polynomial functors can be derived from the subtle interplay between **Poly**^{Cart} and **Poly**.

3.2.1 Epi-Mono Factorization on Poly^{Cart}

In fact, **Poly**^{Cart} itself inherits a factorization system from the epi-mono factorization on types considered previously.

Define a Cartesian lens $(f, f\#) : p \rightleftharpoons q$ to be a *Cartesian embedding* if f is a monomorphism, and a *Cartesian surjection* if f is an epimorphism.

```

isCartesianEmbedding : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → isCartesian p q f → Set (ℓ ⊔ ℓ')
isCartesianEmbedding p q (f, f#) cf = isMono f

isCartesianSurjection : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → isCartesian p q f → Set ℓ'
isCartesianSurjection p q (f, f#) cf = isEpi f

```

Then every Cartesian lens can be factored into a Cartesian surjection followed by a Cartesian embedding.

```

factorcart1 : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → isCartesian p q f
  → p ⇔ (Im (fst f), λ (x, _) → snd q x)
factorcart1 p q (f, f#) cf =
  (factor1 f), f#

factorcart1IsCart : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) (cf : isCartesian p q f)
  → isCartesian p
    (Im (fst f), λ (x, _) → snd q x)
    (factorcart1 p q f cf)
factorcart1IsCart p q (f, f#) cf = cf

factorcart1IsEpi : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) (cf : isCartesian p q f)

```

```

      → isCartesianSurjection p
      (Im (fst f) , λ (x , _) → snd q x)
      (factorcart1 p q f cf)
      (factorcart1IsCart p q f cf)
factorcart1IsEpi p q (f , f#) cf = factor1IsEpi f

factorcart2 : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) → isCartesian p q f
  → (Im (fst f) , λ (x , _) → snd q x) ⇔ q
factorcart2 p q (f , f#) cf = (factor2 f) , λ (x , _) y → y

factorcart2IsCart : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) (cf : isCartesian p q f)
  → isCartesian (Im (fst f) , λ (x , _) → snd q x) q
  (factorcart2 p q f cf)
factorcart2IsCart p q (f , f#) cf x = idIsEquiv

factorcart2IsMono : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) (cf : isCartesian p q f)
  → isCartesianEmbedding
  (Im (fst f) , λ (x , _) → snd q x) q
  (factorcart2 p q f cf)
  (factorcart2IsCart p q f cf)
factorcart2IsMono p q (f , f#) cf = factor2IsMono f

factorcart≡ : ∀ {ℓ ℓ' κ κ'} (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (f : p ⇔ q) (cf : isCartesian p q f)
  → EqLens p q f
  (comp p (Im (fst f) , λ (x , _) → snd q x) q
    (factorcart1 p q f cf)
    (factorcart2 p q f cf))
factorcart≡ p q f cf x = refl , λ y → refl

```

3.3 Composition of Polynomial Functors

As endofunctors on **Type**, polynomial functors may straightforwardly be composed. To show that the resulting composite is itself (equivalent to) a polynomial functor, we can reason via the following chain of equivalences: given polynomials (A , B) and (C , D), their composite, evaluated at a type y is

$$\begin{aligned}
& \sum_{a:A} \prod_{b:B(a)} \sum_{c:C} y^{D(c)} \\
& \simeq \sum_{a:A} \sum_{f:B(a) \rightarrow C} \prod_{b:B(a)} y^{D(f(b))} \\
& \simeq \sum_{(a,f): \sum_{a:A} C^{B(a)}} y^{\sum_{b:B(a)} D(f(b))}
\end{aligned}$$

This then defines a monoidal product \triangleleft on **Poly** with monoidal unit given by the identity functor y :

$$\begin{aligned} _ \triangleleft _ &: \forall \{\ell \ell' \kappa \kappa'\} \rightarrow \text{Poly } \ell \kappa \rightarrow \text{Poly } \ell' \kappa' \rightarrow \text{Poly } (\ell \sqcup \kappa \sqcup \ell') (\kappa \sqcup \kappa') \\ (A, B) \triangleleft (C, D) &= (\Sigma A (\lambda a \rightarrow B a \rightarrow C), \lambda (a, f) \rightarrow \Sigma (B a) (\lambda b \rightarrow D (f b))) \end{aligned}$$

$$\begin{aligned} \triangleleft \text{Lens} &: \forall \{\ell \ell' \ell'' \ell''' \kappa \kappa' \kappa'' \kappa'''\} \\ &\rightarrow (p : \text{Poly } \ell \kappa) (p' : \text{Poly } \ell' \kappa') \\ &\rightarrow (q : \text{Poly } \ell'' \kappa'') (q' : \text{Poly } \ell''' \kappa''') \\ &\rightarrow p \rightleftharpoons p' \rightarrow q \rightleftharpoons q' \rightarrow (p \triangleleft q) \rightleftharpoons (p' \triangleleft q') \\ \triangleleft \text{Lens } p \, p' \, q \, q' \, (f, g) \, (h, k) &= \\ &((\lambda (a, c) \rightarrow (f a, \lambda b' \rightarrow h (c (g a b'))))) \\ &, \lambda (a, c) (b', d') \rightarrow ((g a b'), k (c (g a b')) d')) \end{aligned}$$

where $\triangleleft \text{Lens}$ is the action of \triangleleft on lenses.

By construction, the existence of a Cartesian lens $(\sigma, \sigma\#) : u \triangleleft u \rightleftharpoons u$ effectively shows that u is closed under Σ -types, since:

- σ maps a pair (A, B) consisting of $A : U$ and $B : (u A) \rightarrow U$ to a term $\sigma(A, B)$ representing the Σ type. This corresponds to the type formation rule

$$\frac{\Gamma \vdash A : \text{Type} \quad \Gamma, x : A \vdash B[x] \text{Type}}{\Gamma \vdash \Sigma x : A. B[x] \text{Type}}$$

◇ **David says:** I like the $B[x]$ -bracket notation, but didn't you use $B(x)$ -paren notation above? Or is there a subtle distinction coming up that I've missed? ◇

- For all (A, B) as above, $\sigma\#(A, B)$ takes a term of type $\sigma(A, B)$ and yields a term $\text{fst } (\sigma\#(A, B)) : A$ along with a term $\text{snd } (\sigma\#(A, B)) : B(\text{fst } (\sigma\#(A, B)))$, corresponding to the elimination rules

$$\frac{\Gamma \vdash p : \Sigma x : A. B[x]}{\Gamma \vdash \pi_1(p) : A} \quad \frac{\Gamma \vdash p : \Sigma x : A. B[x]}{\Gamma \vdash \pi_2(p) : B[\pi_1(p)]}$$

- The fact that $\sigma\#(A, B)$ is an equivalence implies it has an inverse $\sigma\#^{-1}(A, B) : \Sigma (u A) (\lambda x \rightarrow u (B x)) \rightarrow u (\sigma(A, B))$, which takes a pair of terms to a term of the corresponding pair type, and thus corresponds to the introduction rule

$$\frac{\Gamma \vdash a : A \quad \Gamma \vdash b : B[a]}{\Gamma \vdash (a, b) : \Sigma x : A. B[x]}$$

- The fact that $\sigma\#^{-1}(A, B)$ is both a left and a right inverse to $\sigma\#$ then implies the usual β and η laws for dependent pair types

$$\pi_1(a, b) = a \quad \pi_2(a, b) = b \quad p = (\pi_1(p), \pi_2(p))$$

Similarly, the existence of a Cartesian lens $(\eta, \eta\#) : y \rightleftharpoons u$ implies that u contains (a type equivalent to) the unit type, in that:

- There is an element $\eta \text{ tt} : U$ which represents the unit type. This corresponds to the type formation rule

$$\frac{}{\Gamma \vdash \top : \text{Type}}$$

- The “elimination rule” $\eta^\# \text{ tt} : u(\eta \text{ tt}) \rightarrow \top$, applied to an element $x : u(\eta \text{ tt})$ is trivial, in that it simply discards x . This corresponds to the fact that, in the ordinary type-theoretic presentation, \top does not have an elimination rule.
- However, since this trivial elimination rule has an inverse $\eta^{\#^{-1}} \text{ tt} : \top \rightarrow u(\eta \text{ tt})$, it follows that there is a (unique) element $\eta^{\#^{-1}} \text{ tt} \text{ tt} : u(\eta \text{ tt})$, which corresponds to the introduction rule for \top :

$$\frac{}{\Gamma \vdash \text{tt} : \top}$$

- Moreover, the uniqueness of this element corresponds to the η -law for \top :

$$\frac{\Gamma \vdash x : \top}{\Gamma \vdash x = \text{tt}}$$

But then, what sorts of laws can we expect Cartesian lenses as above to obey, and is the existence of such a lens all that is needed to ensure that the natural model u has dependent pair types in the original sense of Awodey & Newstead’s definition in terms of Cartesian (pseudo)monads, or is some further data required? And what about Π types, or other type formers? To answer these questions, we will need to study the structure of \triangleleft , along with some closely related functors, in a bit more detail. In particular, we shall see that the structure of \triangleleft as a monoidal product on **Poly** reflects many of the basic identities one expects to hold of Σ types.

For instance, the associativity of \triangleleft corresponds to the associativity of Σ -types,

```

<assoc : ∀ {ℓ ℓ' ℓ'' κ κ' κ''}
  → (p : Poly ℓ κ) (q : Poly ℓ' κ') (r : Poly ℓ'' κ'')
  → ((p < q) < r) ≅ (p < (q < r))

<assoc p q r =
  ((λ ((a , f) , g) → (a , (λ b → (f b , λ d → g (b , d)))))
  , λ ((a , f) , g) (b , (d , x)) → ((b , d) , x))

<assocCart : ∀ {ℓ ℓ' ℓ'' κ κ' κ''}
  → (p : Poly ℓ κ) (q : Poly ℓ' κ') (r : Poly ℓ'' κ'')
  → isCartesian ((p < q) < r) (p < (q < r)) (<assoc p q r)

<assocCart p q r (a , f) =
  Iso→isEquiv ( (λ ((b , d) , x) → b , d , x)
    , ( (λ (b , d , x) → refl)
      , λ ((b , d) , x) → refl))

```

while the left and right unitors of \triangleleft correspond to the fact that \top is both a left and a right unit for Σ -types. ♦ *David says:* It might be good to add the type-theoretic conversion rules here? ♦

```

<unitl : ∀ {ℓ κ} (p : Poly ℓ κ) → (y < p) ≃ p
<unitl p = (λ (tt , a) → a tt) , λ (tt , a) x → tt , x

```

```

<unitlCart : ∀ {ℓ κ} (p : Poly ℓ κ)
  → isCartesian (y < p) p (<unitl p)
<unitlCart p (tt , a) =
  Iso→isEquiv ( (λ (tt , b) → b)
    , (λ b' → refl)
    , (λ b' → refl) )

```

```

<unitr : ∀ {ℓ κ} (p : Poly ℓ κ) → (p < y) ≃ p
<unitr p = (λ (a , f) → a) , λ (a , f) b → b , tt

```

```

<unitrCart : ∀ {ℓ κ} (p : Poly ℓ κ)
  → isCartesian (p < y) p (<unitr p)
<unitrCart p (a , f) =
  Iso→isEquiv ( (λ (b , tt) → b)
    , (λ b → refl)
    , (λ (b , tt) → refl) )

```

In fact, \triangleleft restricts to a monoidal product on $\mathbf{Poly}^{\mathbf{Cart}}$, since the functorial action of \triangleleft on lenses preserves Cartesian lenses:

```

<LensCart : ∀ {ℓ ℓ' ℓ'' ℓ''' κ κ' κ'' κ'''}
  → (p : Poly ℓ κ) (q : Poly ℓ' κ')
  → (r : Poly ℓ'' κ'') (s : Poly ℓ''' κ''')
  → (f : p ≃ q) (g : r ≃ s)
  → isCartesian p q f → isCartesian r s g
  → isCartesian (p < r) (q < s)
    (<Lens p q r s f g)
<LensCart p q r s (f , f#) (g , g#) cf cg (a , h) =
  pairEquiv (f# a) (λ x → g# (h (f# a x)))
    (cf a) (λ x → cg (h (f# a x)))

```

We should expect, then, for these equivalences to be somehow reflected in the structure of a Cartesian lenses $\eta : y \simeq u$ and $\mu : u \triangleleft u \simeq u$. This would be the case, e.g., if the following diagrams in $\mathbf{Poly}^{\mathbf{Cart}}$ were to commute

$$\begin{array}{ccc}
 y \triangleleft u & \xrightarrow{\eta \triangleleft \text{id}} & u \triangleleft u \xleftarrow{\text{id} \triangleleft \eta} u \triangleleft y \\
 \searrow \triangleleft \text{unitl} & \downarrow \mu & \swarrow \triangleleft \text{unitr} \\
 & u &
 \end{array}
 \qquad
 \begin{array}{ccccc}
 (u \triangleleft u) \triangleleft u & \xrightarrow{\triangleleft \text{assoc}} & u \triangleleft (u \triangleleft u) & \xrightarrow{\text{id} \triangleleft \mu} & u \triangleleft u \\
 \mu \triangleleft \text{id} \downarrow & & & & \downarrow \mu \\
 u \triangleleft u & \xrightarrow{\mu} & & & u
 \end{array}$$

One may recognize these as the usual diagrams for a monoid in a monoidal category, hence (since \triangleleft corresponds to composition of polynomial endofunctors) for a *monad* as typically defined. However, because of the higher-categorical structure of types in HoTT,

we should not only ask for these diagrams to commute, but for the cells exhibiting that these diagrams commute to themselves be subject to higher coherences, and so on, giving u not the structure of a (Cartesian) monad, but rather of a (Cartesian) ∞ -*monad*.

Yet demonstrating that u is an ∞ -monad involves specifying a potentially infinite amount of coherence data. Have we therefore traded both the Scylla of equality-up-to-isomorphism and the Charybdis of strictness for an even worse fate of higher coherence hell? The answer to this question, surprisingly, is negative, as there is a way to implicitly derive all of this data from a single axiom, which corresponds to the characteristic axiom of HoTT itself: univalence. To show this, we now introduce the central concept of this paper – that of a *polynomial universe*.

4 Polynomial Universes

◇ David says: Introduce the section ◇

4.1 Univalence

For any polynomial $u = (A, B)$ and elements $a, b : A$, we may define a function that takes a proof of $a \equiv b$ to an equivalence $B\ a \simeq B\ b$.

```
idToEquiv : ∀ {ℓ κ} (p : Poly ℓ κ) (a b : fst p)
  → a ≡ b → Σ (snd p a → snd p b) isEquiv
idToEquiv p a b e =
  transp (snd p) e
  , Iso→isEquiv (transp (snd p) (sym e) , (sym1 e , symr e))
```

We say that a polynomial u is *univalent* if for all $a, b : A$, this function is an equivalence.

```
isUnivalent : ∀ {ℓ κ} → Poly ℓ κ → Type (ℓ ⊔ κ)
isUnivalent (A , B) =
  (a b : A) → isEquiv (idToEquiv (A , B) a b)
```

We call this property of polynomials univalence by analogy with the usual univalence axiom of HoTT. Indeed, the univalence axiom simply states that the polynomial functor $(\text{Type} , \lambda X \rightarrow X)$ is itself univalent.

postulate

```
ua : ∀ {ℓ} → isUnivalent (Type ℓ , λ X → X)
```

A key property of polynomial universes – qua polynomial functors – is that every polynomial universe u is a *subterminal object* in $\mathbf{Poly}^{\text{Cart}}$, i.e. for any other polynomial p , the type of Cartesian lenses $p \leqslant u$ is a proposition, i.e. any two Cartesian lenses with codomain u are equal. ◇ David says: What do you think of saying “i.e. we say a polynomial v is *univalent* if any two Cartesian lenses with codomain v are equal.”

I wasn't sure why you chose to call this a universe below; I'd be tempted to call it a univalent type family and reserve "universe" for univalent polynomials having more structure, e.g. sums and products. \diamond

```
isSubterminal :  $\forall \{ \ell \ \kappa \} \ (u : \text{Poly } \ell \ \kappa) \rightarrow \text{Set } \omega$ 
isSubterminal u =  $\forall \{ \ell' \ \kappa' \} \ (p : \text{Poly } \ell' \ \kappa') \rightarrow$ 
   $\rightarrow (f \ g : p \hookrightarrow u)$ 
   $\rightarrow \text{isCartesian } p \ u \ f$ 
   $\rightarrow \text{isCartesian } p \ u \ g$ 
   $\rightarrow \text{EqLens } p \ u \ f \ g$ 
```

To show this, we first prove the following *transport lemma*, which says that transporting along an identity $a \equiv b$ induced by an equivalence $f : B \ a \simeq B \ b$ in a univalent polynomial $p = (A \ , \ B)$ is equivalent to applying f .

```
transpLemma :  $\forall \{ \ell \ \kappa \} \ \{ u : \text{Poly } \ell \ \kappa \}$ 
   $\rightarrow (\text{univ} : \text{isUnivalent } u)$ 
   $\rightarrow \{ a \ b : \text{fst } u \} \ (f : \text{snd } u \ a \rightarrow \text{snd } u \ b)$ 
   $\rightarrow (\text{ef} : \text{isEquiv } f) \ (x : \text{snd } u \ a)$ 
   $\rightarrow \text{transp } (\text{snd } u) \ (\text{inv } (\text{univ } a \ b) \ (f \ , \ \text{ef})) \ x \equiv f \ x$ 
transpLemma  $\{ u = u \}$  univ  $\{ a = a \} \ \{ b = b \}$  f ef x =
  coAp (ap fst (snd (snd (univ a b)) ((f \ , \ ef)))) x
```

The result then follows:

```
univ $\rightarrow$ Subterminal :  $\forall \{ \ell \ \kappa \} \ (u : \text{Poly } \ell \ \kappa)$ 
   $\rightarrow \text{isUnivalent } u$ 
   $\rightarrow \text{isSubterminal } u$ 
univ $\rightarrow$ Subterminal u univ p f g cf cg a =
  ( inv univfg (fg $^{-1}$  \ , \ efg $^{-1}$ )
  , ( $\lambda \ b \rightarrow \text{sym } ((\text{snd } g \ a \ (\text{transp } (\text{snd } u) \ (\text{inv } \text{univfg } (\text{fg}^{-1} \ , \ \text{efg}^{-1})) \ b))$ 
     $\equiv \langle \text{ap } (\text{snd } g \ a) \ (\text{transpLemma } \text{univ } \text{fg}^{-1} \ \text{efg}^{-1} \ b) \rangle$ 
     $((\text{snd } g \ a \ (\text{fg}^{-1} \ b))$ 
     $\equiv \langle \text{snd } (\text{snd } (\text{cg } a)) \ (\text{snd } f \ a \ b) \rangle$ 
     $((\text{snd } f \ a \ b) \ \square))))$ 
  where univfg : isEquiv (idToEquiv u (fst f a) (fst g a))
    univfg = univ (fst f a) (fst g a)
    fg $^{-1}$  : snd u (fst f a)  $\rightarrow$  snd u (fst g a)
    fg $^{-1}$  x = inv (cg a) (snd f a x)
    efg $^{-1}$  : isEquiv fg $^{-1}$ 
    efg $^{-1}$  = compIsEquiv (inv (cg a)) (snd f a)
      (invIsEquiv (cg a)) (cf a)
```

We shall refer to polynomial functors with this property of being subterminal objects in $\mathbf{Poly}^{\text{Cart}}$ as *polynomial universes*. As we shall see, such polynomial universes have many desirable properties as models of type theory.

If we think of a polynomial p as representing a family of types, then what this tells us is that if u is a polynomial universe, there is essentially at most one way for u to contain the types represented by p , where Containment is here understood as \diamond David says: mere? \diamond existence of a Cartesian lens $p \hookrightarrow u$. In this case, we say that u *classifies* the types represented by p .

As a direct consequence of this fact \diamond David says: which fact? say again, since the last thing discussed was just terminology. \diamond , it follows that every diagram consisting of parallel Cartesian lenses into a polynomial universe automatically commutes, and moreover, every higher diagram that can be formed between the cells exhibiting such commutation must also commute, etc.

Hence, due to the above theorem and the closure of Cartesian lenses under composition, it follows that u *automatically* satisfies the laws of a monad if there are Cartesian lenses $\eta : y \hookrightarrow u$ and $\mu : u \triangleleft u \hookrightarrow u$.

```
univ<unitl : ∀ {ℓ κ} (u : Poly ℓ κ) → isUnivalent u
  → (η : y ↪ u) (μ : (u < u) ↪ u)
  → isCartesian y u η → isCartesian (u < u) u μ
  → EqLens (y < u) u
    (<unitl u)
    (comp (y < u) (u < u) u
      (<Lens y u u u η (id u)) μ)
```

```
univ<unitl u univ η μ cη cμ =
  univ→Subterminal
    u univ (y < u) (<unitl u)
    (comp (y < u) (u < u) u
      (<Lens y u u u η (id u)) μ)
    (<unitlCart u)
    (compCartesian (y < u) (u < u) u
      (<Lens y u u u η (id u)) μ
      (<LensCart y u u u η (id u)
        cη (idCart u)) cμ)
```

```
univ<unitr : ∀ {ℓ κ} (u : Poly ℓ κ) → isUnivalent u
  → (η : y ↪ u) (μ : (u < u) ↪ u)
  → isCartesian y u η → isCartesian (u < u) u μ
  → EqLens (u < y) u
    (<unitr u)
    (comp (u < y) (u < u) u
      (<Lens u u y u (id u) η) μ)
```

```
univ<unitr u univ η μ cη cμ =
  univ→Subterminal
    u univ (u < y) (<unitr u)
    (comp (u < y) (u < u) u
      (<Lens u u y u (id u) η) μ)
```

```

(≪unitrCart u)
(compCartesian (u ≪ y) (u ≪ u) u
  (≪Lens u u y u (id u) η) μ
  (≪LensCart u u y u (id u) η
    (idCart u) cη) cμ)

univ≪assoc : ∀ {ℓ κ} (u : Poly ℓ κ) → isUnivalent u
  → (η : y ≌ u) (μ : (u ≪ u) ≌ u)
  → isCartesian y u η → isCartesian (u ≪ u) u μ
  → EqLens ((u ≪ u) ≪ u) u
    (comp ((u ≪ u) ≪ u) (u ≪ u) u
      (≪Lens (u ≪ u) u u u μ (id u)) μ)
    (comp ((u ≪ u) ≪ u) (u ≪ (u ≪ u)) u
      (≪assoc u u u)
      (comp (u ≪ (u ≪ u)) (u ≪ u) u
        (≪Lens u u (u ≪ u) u
          (id u) μ) μ))

univ≪assoc u univ η μ cη cμ =
  univ→Subterminal
    u univ ((u ≪ u) ≪ u)
    (comp ((u ≪ u) ≪ u) (u ≪ u) u
      (≪Lens (u ≪ u) u u u μ (id u)) μ)
    (comp ((u ≪ u) ≪ u) (u ≪ (u ≪ u)) u
      (≪assoc u u u)
      (comp (u ≪ (u ≪ u)) (u ≪ u) u
        (≪Lens u u (u ≪ u) u (id u) μ) μ))
    (compCartesian ((u ≪ u) ≪ u) (u ≪ u) u
      (≪Lens (u ≪ u) u u u μ (id u)) μ
      (≪LensCart (u ≪ u) u u u μ (id u)
        cμ (idCart u)) cμ)
    (compCartesian ((u ≪ u) ≪ u) (u ≪ (u ≪ u)) u
      (≪assoc u u u)
      (comp (u ≪ (u ≪ u)) (u ≪ u) u
        (≪Lens u u (u ≪ u) u
          (id u) μ) μ)
      (≪assocCart u u u)
      (compCartesian
        (u ≪ (u ≪ u)) (u ≪ u) u
        (≪Lens u u (u ≪ u) u (id u) μ) μ
        (≪LensCart u u (u ≪ u) u (id u) μ
          (idCart u) cμ) cμ))

```

And more generally, all the higher coherences of an ∞ -monad should ◇ David
says: What work is “should” doing here? ◇ follow from the contractibility of the types

of Cartesian lenses $p \rightleftharpoons u$ that can be formed using μ and η .

4.1.1 Rezk Completion of Natural Models

We have so far seen that polynomial universes are quite special objects in the theory of polynomial functors in HoTT, but what good would such special objects do us if they turned out to be exceedingly rare or difficult to construct? Moreover, although we have just demonstrated that any polynomial universe is a natural model in Awodey & Newstead's sense \diamond [David says](#): We did? I missed this. I'm guessing it's on its way :) \diamond , one might be inclined to wonder what can be said about the converse direction – does every natural model give rise to a polynomial universe?

In fact, we can answer this latter question in the affirmative, using a familiar construct from the theory of categories in HoTT – the *Rezk Completion*. In the case of polynomial functors/natural models, we will show that this construction allows us to quotient any polynomial functor to a corresponding polynomial universe, which classifies the unit type and Σ types if the original polynomial does.

By our assumption of the univalence axiom, every polynomial functor p is classified by *some* univalent polynomial:

```
classifier : ∀ {ℓ κ} (p : Poly ℓ κ) → p ≃ (Type κ , λ X → X)
classifier (A , B) = (B , λ a b → b)
```

```
classifierCart : ∀ {ℓ κ} (p : Poly ℓ κ)
  → isCartesian p (Type κ , λ X → X)
  (classifier p)
classifierCart p a = idIsEquiv
```

We then obtain the Rezk completion of p as the image factorization in $\mathbf{Poly}^{\mathbf{Cart}}$ of this classifying lens:

```
Rezk : ∀ {ℓ κ} (p : Poly ℓ κ) → Poly (lsuc κ) κ
Rezk (A , B) = (Im B) , (λ (X , _) → X)
```

```
→Rezk : ∀ {ℓ κ} (p : Poly ℓ κ) → p ≃ (Rezk p)
→Rezk {κ = κ} p =
  factorcart1 p (Type κ , λ X → X)
    (classifier p)
    (classifierCart p)
```

```
Rezk→ : ∀ {ℓ κ} (p : Poly ℓ κ) → (Rezk p) ≃ (Type κ , λ X → X)
Rezk→ {κ = κ} p =
  factorcart2 p (Type κ , λ X → X)
    (classifier p)
    (classifierCart p)
```

Because the map $\text{Rezk} \rightarrow$ defined above is a Cartesian embedding, and the polynomial $(\text{Type } \kappa, \lambda X \rightarrow X)$ is univalent, it follows that $\text{Rezk } p$ is a polynomial universe:

```

RezkSubterminal : ∀ {ℓ κ} (p : Poly ℓ κ) → isSubterminal (Rezk p)
RezkSubterminal {κ = κ} p q (f , f#) (g , g#) cf cg x =
  ( pairEq (inv (ua (fst (f x)) (fst (g x))))
    ( (λ y → inv (cg x) (f# x y))
      , compIsEquiv (inv (cg x))
                    (f# x)
                    (invIsEquiv (cg x))
                    (cf x))) ||-||IsProp
    , λ y → f# x y
    ≡< sym (g# x (transp (λ X → X)
                        (inv (ua (fst (f x)) (fst (g x))))
                        ((λ z → inv (cg x) (f# x z)) , (compIsEquiv (inv (cg x))
                          (invIsEquiv (cg x))
                          (cf x)))) y)
      ≡< (ap (g# x)
        (transpLemma ua
          (λ z → inv (cg x) (f# x z))
          (compIsEquiv (inv (cg x)) (f# x)
            (invIsEquiv (cg x)) (cf x))
          y)) >
      snd (snd (cg x)) (f# x y)) >
    ap (g# x) (sym (lemma1 ||-||IsProp y)) )
where lemma1 : {a b : fst (Rezk p)}
  → {e : fst a ≡ fst b}
  → (e' : transp (λ c → || (Fibre (snd p) c) ||)
    e (snd a)
    ≡ (snd b))
  → (z : fst a)
  → transp (snd (Rezk p)) (pairEq e e') z
    ≡ transp (λ X → X) e z
lemma1 {e = refl} refl z = refl

```

◇ David says: Deal with long line. ◇

5 Π -Types, Jump Monads & Distributive Laws

5.1 The \Uparrow Functor

5.1.1 Jump Morphisms & the Universal Property of \Uparrow

6 Other Type Formers in Polynomial Universes

6.1 Identity Types

6.2 Positive Types

7 Conclusion